

NAG Fortran Library Routine Document

G05MKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G05MKF generates a vector of pseudo-random integers from the discrete Poisson distribution with mean λ .

2 Specification

```
SUBROUTINE G05MKF(MODE, LAMBDA, N, X, IGEN, ISEED, R, NR, IFAIL)
INTEGER          MODE, N, X(N), IGEN, ISEED(4), NR, IFAIL
real           LAMBDA, R(NR)
```

3 Description

G05MKF generates n integers x_i from a discrete Poisson distribution with mean λ , where the probability of $x_i = I$ is

$$P(x_i = I) = \frac{\lambda^I \times e^{-\lambda}}{I!}, \quad I = 0, 1, \dots,$$

where $0 \leq \lambda$.

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05MKF with the same parameter values can then use this reference vector to generate further variates. The reference array is found using a recurrence relation if λ is less than 50 and by Stirling's formula otherwise.

One of the initialisation routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05MKF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine:

MODE = 0

Set up reference vector only.

MODE = 1

Generate variates using reference vector set up in a prior call to G05MKF.

MODE = 2

Set up reference vector and generate variates.

MODE = 3

Generate variates without using the reference vector.

- Constraint:* $0 \leq \text{MODE} \leq 3$.
- 2: LAMBDA – *real* *Input*
On entry: the mean λ of the Poisson distribution.
Constraint: LAMBDA ≥ 0.0 .
- 3: N – INTEGER *Input*
On entry: the number, n , of pseudo-random numbers to be generated.
Constraint: N ≥ 1 .
- 4: X(N) – INTEGER array *Output*
On exit: the n pseudo-random numbers from the specified Poisson distribution.
- 5: IGEN – INTEGER *Input*
On entry: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialisation by a prior call to one of the routines G05KBF or G05KCF.
- 6: ISEED(4) – INTEGER array *Input/Output*
On entry: contains values which define the current state of the selected generator.
On exit: contains updated values defining the new state of the selected generator.
- 7: R(NR) – *real* array *Input/Output*
On exit: the reference vector.
- 8: NR – INTEGER *Input*
On entry: the dimension of the array R as declared in the (sub)program from which G05MKF is called.
Suggested value: $22 + 20 \times \sqrt{\text{LAMBDA}}$.
Constraints:
 if MODE = 0 or 2, then

$$\text{NR} > (\text{INT}[\text{LAMBDA} + 7.15\sqrt{\text{LAMBDA}} + 8.5] \\ - \max(0, \text{INT}[\text{LAMBDA} - 7.15\sqrt{\text{LAMBDA}}]) + 6);$$

 if MODE = 1, then NR should remain unchanged from the previous call to G05MKF;
 if MODE = 3, then R is not referenced.
- 9: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $N < 1$.

$IFAIL = 2$

On entry, NR is too small when $MODE = 0$ or 2 (see Section 5).

$IFAIL = 3$

On entry, $LAMBDA < 0.0$.

$IFAIL = 4$

On entry, $MODE < 0$
or $MODE > 3$.

$IFAIL = 5$

$MODE = 0$ or 2 and $LAMBDA$ is such that NR would have to be larger than the largest representable integer. Use $MODE = 3$ in this case.

$IFAIL = 6$

$LAMBDA$ is not the same as when R was set up in a previous call with $MODE = 0$ or 2 .

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

The example program prints 10 pseudo-random integers from a Poisson distribution with mean $\lambda = 20$, generated by a single call to G05MKF, after initialisation by G05KBF.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G05MKF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
INTEGER          NOUT, N, NR
PARAMETER       (NOUT=6,N=10,NR=120)
*      .. Local Scalars ..
real           LAMBDA
INTEGER          I, IFAIL, IGEN
*      .. Local Arrays ..
real           R(NR)
INTEGER          ISEED(4), X(N)
*      .. External Subroutines ..
EXTERNAL        G05KBF, G05MKF
```

```
* .. Executable Statements ..
WRITE (NOUT,*) 'G05MKF Example Program Results'
WRITE (NOUT,*)
* Set the distribution parameter LAMBDA
LAMBDA = 20.0e0
* Initialise the seed to a repeatable sequence
ISEED(1) = 1762543
ISEED(2) = 9324783
ISEED(3) = 42344
ISEED(4) = 742355
* IGEN identifies the stream.
IGEN = 1
CALL G05KBF(IGEN,ISEED)
IFAIL = 0
* Generate reference vector R
CALL G05MKF(0,LAMBDA,N,X,IGEN,ISEED,R,NR,IFAIL)
* Generate integers and store in X
CALL G05MKF(1,LAMBDA,N,X,IGEN,ISEED,R,NR,IFAIL)
*
WRITE (NOUT,99999) (X(I),I=1,N)
STOP
*
99999 FORMAT (1X,I12)
END
```

9.2 Program Data

None.

9.3 Program Results

G05MKF Example Program Results

```
14
28
19
23
28
19
21
19
18
26
```
